query: #title: Most frequent P734 values without property (Germany, POB)
  # Complex constraint from [[Property talk:P6597]]
  # family names used as P734 values, but without a P6597 statement. Selection by place of birth
  SELECT ?item ?l ?count ?sample
  WITH
  {
    SELECT ?item (COUNT(DISTINCT ?p) as ?count) (SAMPLE(?p) as ?sample)
    {
      ?p wdt:P19 / wdt:P17 wd:Q183 .
      hint:Prior hint:rangeSafe true .    
      ?p wdt:P734 ?item .
      ?p wdt:P31 wd:Q5 .
    }
    GROUP BY ?item
    HAVING ( ?count > 20 )           
  } as %a         
  WHERE
  {
    INCLUDE %a
    FILTER NOT EXISTS { ?item wdt:P6597 [] }
    ?item rdfs:label ?l . FILTER(lang(?l) = "de" ) 
  }
  ORDER BY DESC(?count)
  LIMIT 100